Constrained minimization using Lagrange multipliers

Function

```
A = {{3, 2}, {2, 6}};
ϕ[x_, y_] := 0.5 {x, y}.A.{x, y} - {x, y}.{3, -1}
```

```
ϕ[x, y] // Expand
```

$-3\,x + 1.5\,x^2 + y + 2.\,x\,y + 3.\,y^2$

```
g1 = Plot3D[ϕ[x, y], {x, -5, 5}, {y, -5, 5}]
```



```
ContourPlot[ϕ[x, y], {x, -5, 5}, {y, -5, 5}]
```



Global minimum

```
Minimize[ϕ[x, y], {x, y}]
```

```
{-2.464285714, {x → 1.428571429, y → -0.6428571429}}
```

```
xmin = Minimize[ϕ[x, y], {x, y}][[2]]
```

```
{x → 1.428571429, y → -0.6428571429}
```

```
g2 = ListPointPlot3D[{{x, y, ϕ[x, y]}} /. xmin]
```



```
Show[g1, g2]
```



Constraint

```
h1 = ParametricPlot3D[{Cos[θ], Sin[θ], 0}, {θ, -Pi, Pi}]
```



```
g3 = ParametricPlot3D[{Cos[θ], Sin[θ], ϕ[Cos[θ], Sin[θ]]}, {θ, -Pi, Pi}]
```




```
Show[g3, h1]
```



```
Show[g1, g3, g2]
```



Contrained minimization (without Lagrage multipliers)

```
Minimize[ϕ[Cos[θ], Sin[θ]], θ]
```

```
{-2.142790959, {θ → -0.408647471}}
```

```
θmin = Minimize[ϕ[Cos[θ], Sin[θ]], θ][[2]]
```

```
{θ → -0.408647471}
```

```
pmin = ListPointPlot3D[
  {{Cos[θ], Sin[θ], ϕ[Cos[θ], Sin[θ]]}} /. θmin, PlotStyle → {Red, Thick}]
```



```
Show[g1, g3, g2, pmin]
```



Contrained minimization (with Lagrage multipliers)

```
D[x^2 + y^2, {{x, y}}]
```

```
{2 x, 2 y}
```

```
g[x_, y_] := x^2 + y^2
```

```
D[ϕ[x, y] + λ g[x, y], {{x, y}}]
```

```
{-3 + 0.5 (6 x + 4 y) + 2 x λ, 1 + 0.5 (4 x + 12 y) + 2 y λ}
```

```
Solve[{D[ϕ[x, y] + λ g[x, y], {{x, y}}] == 0, g[x, y] == 1}, {x, y, λ}]
```

Solve : Solve was unable to solve the system with inexact coefficients. The answer was obtained by solving a corresponding exact system and numericizing the result.

```
{{x → -0.8742676594, y → -0.485444188, λ → -3.770979793},
 {x → -0.7518019885, y → 0.6593889369, λ → -2.618128278},
 {x → -0.3315894685, y → 0.9434237777, λ → -3.17850995},
 {x → 0.9176591165, y → -0.3973685266, λ → 0.5676180212}}
```

```
{Cos[θ], Sin[θ]} /. θmin
```

```
{0.9176591144, -0.3973685313}
```

We found the minimum (together with the other critical points) using the Lagrange multipliers method!

Homework

1. Reproduce the intuitive explanation (better if you can explain it to

someone)
2. Try to find another characterization of the constrained minimum (with parametrization ?)
3. Ideas o Quantum Chemistry, Piela, Appendix on Variational Principle
4. Can you come up with an intuitive explanation, on R3 as the domain (e.g., temperature function), of a constrained minimization?
5. Can you extend your intuitive explanation to two constraints?(Why the gradient of the function is a linear combination of the gradient of the two constraints!?)